(*
  Copyright (c) 2012, Stefan Glienke
  All rights reserved.

  Redistribution and use in source and binary forms, with or without
  modification, are permitted provided that the following conditions are met:

  - Redistributions of source code must retain the above copyright notice,
    this list of conditions and the following disclaimer.
  - Redistributions in binary form must reproduce the above copyright notice,
    this list of conditions and the following disclaimer in the documentation
    and/or other materials provided with the distribution.
  - Neither the name of this library nor the names of its contributors may be
    used to endorse or promote products derived from this software without
    specific prior written permission.

  THIS SOFTWARE IS PROVIDED BY THE COPYRIGHT HOLDERS AND CONTRIBUTORS "AS IS"
  AND ANY EXPRESS OR IMPLIED WARRANTIES, INCLUDING, BUT NOT LIMITED TO, THE
  IMPLIED WARRANTIES OF MERCHANTABILITY AND FITNESS FOR A PARTICULAR PURPOSE
  ARE DISCLAIMED. IN NO EVENT SHALL THE COPYRIGHT HOLDER OR CONTRIBUTORS BE
  LIABLE FOR ANY DIRECT, INDIRECT, INCIDENTAL, SPECIAL, EXEMPLARY, OR
  CONSEQUENTIAL DAMAGES (INCLUDING, BUT NOT LIMITED TO, PROCUREMENT OF
  SUBSTITUTE GOODS OR SERVICES; LOSS OF USE, DATA, OR PROFITS; OR BUSINESS
  INTERRUPTION) HOWEVER CAUSED AND ON ANY THEORY OF LIABILITY, WHETHER IN
  CONTRACT, STRICT LIABILITY, OR TORT (INCLUDING NEGLIGENCE OR OTHERWISE)
  ARISING IN ANY WAY OUT OF THE USE OF THIS SOFTWARE, EVEN IF ADVISED OF THE
  POSSIBILITY OF SUCH DAMAGE.
*)

unit DSharp.Interception.InterceptionBehaviorPipeline;

interface

uses
  DSharp.Interception,
  Generics.Collections;

type
  TInterceptionBehaviorPipeline = class
  private
    FInterceptionBehaviors: TList<IInterceptionBehavior>;
    function GetCount: Integer;
  public
    constructor Create; overload;
    constructor Create(InterceptionBehaviors: array of IInterceptionBehavior); overload;
    destructor Destroy; override;

    procedure Add(InterceptionBehavior: IInterceptionBehavior);
    function Invoke(Input: IMethodInvocation; Target: TInvokeInterceptionBehaviorDelegate): IMethodReturn;

    property Count: Integer read GetCount;
  end;

implementation

{ TInterceptionBehaviorPipeline }

constructor TInterceptionBehaviorPipeline.Create;
begin
  FInterceptionBehaviors := TList<IInterceptionBehavior>.Create;
end;

constructor TInterceptionBehaviorPipeline.Create(
  InterceptionBehaviors: array of IInterceptionBehavior);
begin
  Create;
  FInterceptionBehaviors.AddRange(InterceptionBehaviors);
end;

destructor TInterceptionBehaviorPipeline.Destroy;
begin
  FInterceptionBehaviors.Free;
  inherited;
end;

function TInterceptionBehaviorPipeline.GetCount: Integer;
begin
  Result := FInterceptionBehaviors.Count;
end;

procedure TInterceptionBehaviorPipeline.Add(
  InterceptionBehavior: IInterceptionBehavior);
begin
  FInterceptionBehaviors.Add(InterceptionBehavior);
end;

function TInterceptionBehaviorPipeline.Invoke(Input: IMethodInvocation;
  Target: TInvokeInterceptionBehaviorDelegate): IMethodReturn;
var
  interceptorIndex: Integer;
begin
  if FInterceptionBehaviors.Count = 0 then
  begin
    Result := Target(Input, nil);
  end
  else
  begin
    interceptorIndex := 0;

    Result := FInterceptionBehaviors[0].Invoke(Input,
      function: TInvokeInterceptionBehaviorDelegate
      begin
        Inc(interceptorIndex);
        if interceptorIndex < FInterceptionBehaviors.Count then
        begin
          Result := FInterceptionBehaviors[interceptorIndex] as IInvokeInterceptionBehaviorDelegate;
        end
        else
        begin
          Result := Target;
        end;
      end);
  end;
end;

end.
